# $Id: PKGBUILD 184417 2013-05-06 19:39:02Z foutrelis $
# Maintainer:
# Contributor: Ronald van Haren <ronald.archlinux.org>
# Contributor: Andries Radu <admiral0@live.it>

pkgname=opengtl
pkgver=0.9.18
pkgrel=2
pkgdesc="A set of libraries for using and integrating transformation algorithms (such as filter or color conversion) in graphics applications"
url="http://www.opengtl.org"
arch=('i686' 'x86_64')
license=('GPL')
depends=('gcc-libs' 'llvm')
makedepends=('cmake' 'libpng')
optdepends=('libpng: for using the png extension')
source=(http://download.opengtl.org/OpenGTL-$pkgver.tar.bz2
        opengtl-0.9.18-memcpy.patch)
sha256sums=('f094b3d2a1a60519975bb3badc05e704f2b93c7a3d2cb753041f2ff27cfcd9f8'
            '357cc2e538bbaaffe7d58956959d8a7fbae09741d4a87488a5a8c51ec645d23a')

prepare() {
  mkdir build
  patch -d OpenGTL-$pkgver -Np1 -i "$srcdir/opengtl-0.9.18-memcpy.patch"
}

build() {
  cd "$srcdir/build"
  cmake ../OpenGTL-$pkgver \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DCMAKE_BUILD_TYPE=Release \
    -DCMAKE_SKIP_RPATH=ON
  make
}

package() {
  cd "$srcdir/build"
  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
